/*
 *  Copyright (c) 2010 Brian Stanek <bstanek@gmail.com>
 *
 *  Permission is hereby granted, free of charge, to any person
 *  obtaining a copy of this software and associated documentation
 *  files (the "Software"), to deal in the Software without
 *  restriction, including without limitation the rights to use,
 *  copy, modify, merge, publish, distribute, sublicense, and/or sell
 *  copies of the Software, and to permit persons to whom the
 *  Software is furnished to do so, subject to the following
 *  conditions:
 *
 *  The above copyright notice and this permission notice shall be
 *  included in all copies or substantial portions of the Software.
 *
 *  THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND,
 *  EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES
 *  OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND
 *  NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT
 *  HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY,
 *  WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING
 *  FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR
 *  OTHER DEALINGS IN THE SOFTWARE.
 */

package guice.scopes
{

import guice.Scope;
import guice.Stage;
import guice.binder.ScopedBindingBuilder;
import guice.spi.BindingScopingVisitor;

public class Scoping
{

    //--------------------------------------------------------------------------
    //
    //  Class Constants
    //
    //--------------------------------------------------------------------------

    public static const UNSCOPED:Scoping = new UnscopedScoping();

    public static const SINGLETON_INSTANCE:Scoping = new InstanceScoping(Scopes.SINGLETON);

    public static const EAGER_SINGLETON:Scoping = new EagerSingletonScoping();

    //--------------------------------------------------------------------------
    //
    //  Class Methods
    //
    //--------------------------------------------------------------------------

    public static function forInstance(scope:Scope):Scoping
    {
        if (scope == Scopes.SINGLETON)
        {
            return SINGLETON_INSTANCE;
        }
        return new InstanceScoping(scope);
    }

    //--------------------------------------------------------------------------
    //
    //  Abstract Methods
    //
    //--------------------------------------------------------------------------

    public function acceptVisitor(visitor:BindingScopingVisitor):Boolean
    {
        return false;
    }

    public function applyTo(scopedBindingBuilder:ScopedBindingBuilder):void
    {
    }

    //--------------------------------------------------------------------------
    //
    //  Methods
    //
    //--------------------------------------------------------------------------

    public function isExplicitlyScoped():Boolean
    {
        return this != UNSCOPED;
    }

    public function isNoScope():Boolean
    {
        return getScopeInstance() == Scopes.NO_SCOPE;
    }

    public function isEagerSingleton(stage:Stage):Boolean
    {
        if (this == EAGER_SINGLETON)
        {
            return true;
        }

        if (stage == Stage.PRODUCTION)
        {
            return this == SINGLETON_INSTANCE;
        }

        return false;
    }

    public function getScopeInstance():Scope
    {
        return null;
    }

    public function getScopeAnnotation():String
    {
        return null;
    }

}

}
